Results of the spatial BYM model proposed by Mitzi

1 Implementation of the BYM model proposed by Mitzi.

```
data {
  int<lower=0> N;
  int<lower=0> N_edges;
  int<lower=1, upper=N> node1[N_edges];  // node1[i] adjacent to node2[i]
  int<lower=1, upper=N> node2[N_edges];  // and node1[i] < node2[i]

  int<lower=0> y[N];              // count outcomes
//  vector[N] x;                    // predictor
  vector<lower=0>[N] E;           // exposure
}
transformed data {
  vector[N] log_E = log(E);
}
parameters {
  real beta0;                // intercept
//  real beta1;                // slope

  real<lower=0> tau_theta;   // precision of heterogeneous effects
  real<lower=0> tau_phi;     // precision of spatial effects

  vector[N] theta_std;       // standardized heterogeneous effects
  vector[N - 1] phi_std_raw; // raw, standardized spatial effects
}
transformed parameters {
  real<lower=0> sigma_theta = inv(sqrt(tau_theta));  // convert precision to sigma
  vector[N] theta = theta_std * sigma_theta;         // non-centered parameterization

  real<lower=0> sigma_phi = inv(sqrt(tau_phi));      // convert precision to sigma
  vector[N] phi;
  phi[1:(N - 1)] = phi_std_raw;
  phi[N] = -sum(phi_std_raw);
  phi = phi * sigma_phi;    // non-centered parameterization
}
model {
//  y ~ poisson_log(log_E + beta0 + beta1 * x + theta + phi);

  y ~ poisson_log(log_E + beta0 + theta + phi);

  target += -0.5 * dot_self(phi[node1] - phi[node2]);

  beta0 ~ normal(0, 5);
//  beta1 ~ normal(0, 5);
  theta_std ~ normal(0, 1);
  tau_theta ~ gamma(3.2761, 1.81);  // Carlin WinBUGS priors
  tau_phi ~ gamma(1, 1);            // Carlin WinBUGS priors
}
generated quantities {
```

```
//  vector[N] mu = exp(log_E + beta0 + beta1 * x + phi + theta);
  vector[N] mu = exp(log_E + beta0 + phi + theta);
  vector[N] SMR = exp(beta0 + phi + theta);
  real psi = sd(phi) / (sd(theta) + sd(phi));  // proportion spatial variation
}
```

2 Fitting the model to Spain's mortality data.

```
library(rstan)
library(spdep)
# options(mc.cores = parallel::detectCores())
options(mc.cores = 3)

source("mungeCARdata4stan.R")
load("../../Mortalidad nacional/DatosSinTemporal.Rdata")
Veci <- nb2WB(CartoMuniSinIslas.nb)
nbs = mungeCARdata4stan(Veci$adj, Veci$num)
N = nbs$N
node1 = nbs$node1
node2 = nbs$node2
N_edges = nbs$N_edges

tcomp02 <- system.time(mod.BYM <- stan_model("BYM_Mitzi.stan"))
# Compilation in 65.6 seconds

save(mod.BYM, file = "mod.Stan.BYM_Mitzi.Rdata")
# load('mod.Stan.BYM_Mitzi.Rdata')

Ejecuta.BYM.Stan <- function(Sexo, Causa) {
    y <- MorTabu[Sexo, Causa, ]
    E <- Esperados[Sexo, Causa, ]

    datos <- list(N, N_edges, node1, node2, y, E)
    param <- c("SMR", "mu", "beta0", "sigma_phi", "tau_phi", "sigma_theta",
        "tau_theta")
    Res.t <- system.time(Res <- sampling(object = mod.BYM, data = datos, chains = 3,
        iter = 4400, warmup = 400, thin = max(1, floor(3 * (4400 - 400)/1000)),
        cores = 3, pars = param))

    Res.BYM[[Sexo]][[Causa]] <<- list()
    Res.BYM[[Sexo]][[Causa]]$tiempo <<- Res.t
    Res.BYM[[Sexo]][[Causa]]$summary <<- Res
    Res.BYM[[Sexo]][[Causa]]$RR <<- summary(Res)$summary[1:7907, 1]
    Res.BYM[[Sexo]][[Causa]]$P.RR <<- apply(extract(Res, pars = "SMR")$SMR,
        2, function(x) {
            mean(x > 1)
        })
}

Res.BYM <- list(Hombres = list(), Mujeres = list())

for (i in c(9:18, 22, 28, 30, 31, 33, 35:37, 41)) {
```

```
    Ejecuta.BYM.Stan(1, i)
    save(Res.BYM, file = "Res.BYM-StanMitzi.Rdata")
}
```

3 Descriptive principals.

Cause	Time	sigma_phi	sigma_theta	max.Rhat	min.n_eff
(9) Mouth and pharynx	9285.2	0.0159 [0.0157,0.0162]	0.1848 [0.1679,0.2043]	1.11	23.72
(10) Esophagus	5034.2	0.0159 [0.0157,0.0162]	0.1849 [0.1675,0.2026]	1.01	554.85
(11) Stomach	4912.5	0.0159 [0.0157,0.0162]	0.1519 [0.141,0.1647]	1.02	511.89
(12) Colon	9292.6	0.0159 [0.0156,0.0161]	0.1499 [0.1398,0.1606]	1.02	179.91
(13) Rectum	9303.1	0.0159 [0.0156,0.0162]	0.1803 [0.1644,0.1972]	1.04	58.69
(14) Liver	8022.6	0.0159 [0.0157,0.0162]	0.1725 [0.1565,0.1884]	1.01	489.01
(15) Pancreas	5177.2	0.0159 [0.0157,0.0161]	0.1672 [0.1535,0.1824]	1.01	498.12
(16) Other digestives	5001.2	0.0159 [0.0157,0.0162]	0.1916 [0.1737,0.2115]	1.01	472.93
(17) Larynx	4956.3	0.0159 [0.0157,0.0162]	0.1843 [0.1669,0.2033]	1.01	400.92
(18) Lung	4796.9	0.0751 [0.0157,0.2046]	0.2445 [0.1255,0.4664]	89.70	1.50
(22) Other skin	4818.5	0.0159 [0.0156,0.0162]	0.2211 [0.1963,0.2471]	1.02	549.46
(28) Prostate	4797.4	0.0159 [0.0156,0.0162]	0.1415 [0.1314,0.1527]	1.01	534.55
(30) Kidney	9169.2	0.0159 [0.0157,0.0162]	0.1942 [0.1752,0.2136]	1.01	500.50
(31) Bladder	4847.4	0.4053 [0.0157,1.1875]	0.3029 [0.1471,0.5909]	455.78	1.50
(33) Brain	4920.0	0.0159 [0.0157,0.0162]	0.1869 [0.1694,0.2059]	1.01	519.94
(35) Poorly defined	4939.5	0.0159 [0.0157,0.0162]	0.1577 [0.1457,0.1708]	1.01	478.69
(36) Other lymphatics	9117.1	0.0173 [0.0157,0.0245]	0.1713 [0.1558,0.1888]	4.77	1.63
(37) Leukemias	4497.8	0.0159 [0.0156,0.0162]	0.1778 [0.1622,0.1958]	1.02	578.19
(41) Other tumors	8483.7	0.0159 [0.0157,0.0161]	0.1851 [0.1681,0.2043]	1.12	6.84
Median	5001.2	0.02	0.18	1.02	478.69